Excel & Pivot Tables — Complete Reference

EY Probation Training — Week 1. This is the topic to over-prepare: everything else on the roadmap is closer to your existing work.

---

1. Absolute vs Relative References

The single most common source of wrong answers in Excel assessments.

Form	Meaning	On copy
`A1`	Relative	Both row and column shift
`$A$1`	Absolute	Nothing shifts
`A$1`	Row locked	Column shifts, row fixed
`$A1`	Column locked	Row shifts, column fixed

`F4` cycles through all four while editing. Press it repeatedly on a selected reference.

When you need each: a lookup table range must be `$A$2:$D$100` so it doesn't drift as you fill down. A multiplication table needs `A$1` across and `$A1` down. Getting this wrong produces answers that are right in the first cell and wrong everywhere below it — which is exactly what graders look for.

2. Core Functions

Logical

```
=IF(condition, value_if_true, value_if_false)
=IF(AND(A2>100, B2="West"), "Yes", "No")
=IF(OR(A2="X", A2="Y"), 1, 0)
=IFS(A2>90,"A", A2>75,"B", A2>60,"C", TRUE,"D")   ' avoids nested IF
=IFERROR(formula, "fallback")                      ' wrap the WHOLE formula
=IFNA(formula, "not found")                        ' only catches #N/A
```

Aggregation with conditions

```
=SUMIF(range, criteria, [sum_range])
=SUMIFS(sum_range, crit_range1, crit1, crit_range2, crit2, ...)
=COUNTIF(range, criteria)
=COUNTIFS(crit_range1, crit1, crit_range2, crit2, ...)
=AVERAGEIFS(avg_range, crit_range1, crit1, ...)
=MAXIFS(max_range, crit_range1, crit1, ...)
=MINIFS(min_range, crit_range1, crit1, ...)
```

Argument order trap: `SUMIF` puts the sum range LAST; `SUMIFS` puts it FIRST. This inconsistency is deliberately tested.

Criteria syntax — comparison operators go inside quotes:

```
=SUMIFS(C:C, B:B, ">100")            ' greater than 100
=SUMIFS(C:C, B:B, ">="&E1)           ' compare to a cell: concatenate
=COUNTIF(A:A, "*west*")              ' wildcard: contains "west"
=COUNTIF(A:A, "?at")                 ' single-char wildcard: cat, bat, hat
=COUNTIFS(A:A, "<>"&"")              ' non-blank
```

Text

```
=TRIM(text)                    ' removes leading/trailing AND doubled inner spaces
=CLEAN(text)                   ' strips non-printable characters
=UPPER / LOWER / PROPER(text)
=LEFT(text, n)  =RIGHT(text, n)  =MID(text, start, n)
=LEN(text)
=FIND("x", text)               ' case-SENSITIVE, errors if not found
=SEARCH("x", text)             ' case-INsensitive, allows wildcards
=SUBSTITUTE(text, old, new, [instance])
=REPLACE(text, start, n_chars, new)
=CONCAT(A1, " ", B1)   or   =A1&" "&B1
=TEXTJOIN(", ", TRUE, A1:A10)  ' TRUE = skip blanks
=TEXT(value, "0.00%")          ' format a number as text
```

`TRIM` + `UPPER` is the standard cleanup combination for dirty categorical columns.

Dates

```
=TODAY()  =NOW()
=YEAR(d)  =MONTH(d)  =DAY(d)
=EOMONTH(d, 0)                  ' last day of d's month; 0 = current, -1 = previous
=EDATE(d, 3)                    ' same day, 3 months later
=DATEDIF(start, end, "Y")       ' "Y" years, "M" months, "D" days
=NETWORKDAYS(start, end, [holidays])
=WEEKDAY(d, 2)                  ' 2 = Monday is 1
=DATE(year, month, day)
```

Quarter label: `=YEAR(A2)&"-Q"&ROUNDUP(MONTH(A2)/3,0)`

Dates are serial numbers (1 = 1 Jan 1900). Subtracting two dates gives days. If a date shows as a number, it's a formatting issue, not a data issue.

Math and rounding

```
=ROUND(x, 2)   =ROUNDUP(x, 0)   =ROUNDDOWN(x, 0)
=MROUND(x, 5)                   ' nearest multiple of 5
=INT(x)        =TRUNC(x, 2)
=ABS(x)  =MOD(a, b)  =POWER(a, b)  =SQRT(x)
=SUMPRODUCT(range1, range2, ...)
=SUBTOTAL(9, range)             ' 9 = SUM, ignores filtered-out rows
=AGGREGATE(9, 5, range)         ' like SUBTOTAL but can also ignore errors
```

> **SUBTOTAL vs SUM:** `SUM` includes hidden rows; `SUBTOTAL(9,...)` respects the filter. On a filtered dashboard you almost always want `SUBTOTAL`. Function codes: 1=AVERAGE, 2=COUNT, 3=COUNTA, 4=MAX, 5=MIN, 9=SUM. Add 100 (101, 109...) to also ignore manually hidden rows.

3. Lookups

VLOOKUP

```
=VLOOKUP(lookup_value, table_array, col_index, [range_lookup])
=VLOOKUP($A2, Products!$A$2:$F$60, 3, FALSE)
```

- **Always pass FALSE** (exact match). `TRUE` assumes the table is sorted ascending and silently returns approximate matches.
- Lookup column must be the **leftmost** column of `table_array`.
- `col_index` is a hard-coded number — insert a column in the source and the formula silently returns the wrong field. This is the main argument against VLOOKUP.

INDEX + MATCH (preferred)

```
=INDEX(return_range, MATCH(lookup_value, lookup_range, 0))
=INDEX(Products!$C:$C, MATCH($A2, Products!$A:$A, 0))
```

- `MATCH` third argument: `0` = exact (use this), `1` = largest ≤ (needs ascending), `-1` = smallest ≥ (needs descending).
- Works right-to-left, survives column insertion, and reads only the two columns it needs.

Two-way lookup (row and column):

```
=INDEX(data, MATCH(row_val, row_hdrs, 0), MATCH(col_val, col_hdrs, 0))
```

Multi-criteria lookup without array formulas:

```
=INDEX(C:C, MATCH(1, (A:A=E1)*(B:B=E2), 0))          ' Ctrl+Shift+Enter in older Excel
=SUMPRODUCT((A2:A100=E1)*(B2:B100=E2)*C2:C100)       ' if the result is numeric
```

XLOOKUP (Excel 365 / 2021+)

```
=XLOOKUP(lookup, lookup_array, return_array, [if_not_found], [match_mode], [search_mode])
=XLOOKUP($A2, Products!$A:$A, Products!$C:$C, "Not found")
```

Exact match is the DEFAULT. Built-in not-found handling. Can return a whole row/column. `search_mode = -1` searches bottom-up (useful for "most recent" lookups).

> **Assessment tip:** if the machine has Excel 365, use XLOOKUP and say why. If you're unsure of the version, INDEX/MATCH always works. Never lead with VLOOKUP unless asked for it specifically.

4. Data Cleaning Workflow

Before any analysis, spend two minutes on reconnaissance:

1. `Ctrl+End` — find the true extent of the data. Reveals stray cells far below the visible range.
2. `Ctrl+Shift+L` — turn on filters; open each key column's dropdown to see actual distinct values. Casing and whitespace variants show up

immediately.
3. Check for blanks: `=COUNTBLANK(range)`, or filter to Blanks.
4. Check for duplicates: **Data > Remove Duplicates**, or flag with `=COUNTIF($A$2:$A$1000, A2)>1`.

Cleaning tools:

Task	Method
Trim whitespace	`=TRIM(A2)` then paste-special as values
Fix casing	`=PROPER(A2)` / `=UPPER(A2)`
Split a column	Data > Text to Columns (delimited or fixed width)
Merge columns	`=A2&" "&B2` or Flash Fill (`Ctrl+E`)
Remove duplicates	Data > Remove Duplicates (select which columns define a dupe)
Find/replace	`Ctrl+H`; tick "Match entire cell contents" when needed
Convert text-numbers	`=VALUE(A2)`, or Text to Columns > Finish
Fill blanks	`Ctrl+G` > Special > Blanks, type value, `Ctrl+Enter`

Flash Fill (`Ctrl+E`) infers a pattern from one or two examples. Fast for splitting names or reformatting IDs, but it produces static values, not formulas — it won't update when the source changes.

Data Validation (Data > Data Validation): restrict entry to a list, a number range, or a date range. The List source can be a range (`=$H$2:$H$5`) or a typed comma-separated list. This is how you build a dropdown for an interactive summary cell.

5. Pivot Tables

Creating one

`Alt + N + V` (or Insert > PivotTable). **Convert your source to a Table first** (`Ctrl+T`) — then the pivot's source expands automatically when rows are added. Otherwise the source is a fixed range and new rows are silently excluded on refresh.

The four zones

Zone	What it does
Rows	One row per distinct value (nest multiple fields for hierarchy)
Columns	One column per distinct value — keep low-cardinality
Values	The aggregation (Sum, Count, Average, Max, Min, StdDev…)
Filters	Page-level filter above the pivot

Why "Count of Revenue" appears instead of "Sum"

Two causes, and this is a favourite exam question: 1. The column contains **blanks** or **text** anywhere in the range — Excel defaults to Count when the field isn't purely numeric. 2. The numbers are stored **as text** (left-aligned, possibly with a green triangle).

Fix the data, refresh, then right-click the value field > Summarize Values By > Sum. Or double-click the field header to open Value Field Settings.

Show Values As

Right-click a value > **Show Values As**: - `% of Grand Total`, `% of Column Total`, `% of Row Total` - `% of Parent Row Total` — useful with nested row fields - `Difference From` / `% Difference From` — pick a base field and base item (e.g. previous month) - `Running Total In` - `Rank Largest to Smallest`

Grouping

- **Dates:** right-click any date in the pivot > Group > select Years, Quarters, Months (multi-select to nest). Excel creates the hierarchy automatically — you don't need helper columns.
- **Numbers:** right-click > Group > set Starting at, Ending at, By. This is how you bucket ShipDays into 1–3, 4–6, 7–9, 10+.
- **Text:** select several row items > right-click > Group to make a manual grouping.

Calculated Field vs Calculated Item

Crucial distinction, frequently tested.

	Calculated Field	Calculated Item
Operates on	Whole **fields** (columns)	Individual **items** within one field
Appears as	A new field in the Values area	A new row/column within an existing field
Example	`Margin = Revenue - Cost`	`Q1+Q2` as a new item inside the Quarter field
Created via	PivotTable Analyze > Fields, Items & Sets > Calculated Field	…> Calculated Item (select an item in the field first)

> **The big gotcha:** a Calculated Field computes on the **sum of the underlying data**, not on the sum of the displayed values. So `Margin% = (Revenue-Cost)/Revenue` gives `SUM(Revenue-Cost)/SUM(Revenue)` — usually what you want. But a field like `Avg Price = Revenue/Units` returns the ratio of totals, not the average of ratios. Be able to explain this.

Slicers and Timelines

- **Slicer:** PivotTable Analyze > Insert Slicer. Visual filter buttons. One slicer can control **multiple pivots** via Report Connections (right-click slicer > Report Connections) — essential for dashboards.
- **Timeline:** Insert Timeline. Date-only slicer with Years/Quarters/Months/Days granularity.

Other essentials

- **Refresh:** `Alt+F5` for one pivot, `Ctrl+Alt+F5` for all. Pivots do **not** auto-refresh when the source changes.
- **Show Details:** double-click any value cell — Excel generates a new sheet with the underlying rows. Great for auditing an unexpected number.
- **GETPIVOTDATA:** typing `=` and clicking a pivot cell auto-generates this. It's robust to the pivot's layout changing, unlike a plain cell reference.

  ```
  =GETPIVOTDATA("Revenue", $A$3, "Region", "West", "Quarter", "2025-Q4")
  ```

  To turn the auto-generation off: PivotTable Analyze > Options dropdown > untick Generate GetPivotData.
- **Value Filters:** right-click a row label > Filter > Top 10 (works for any N, and for Top N by any value field).
- **Repeat item labels:** Design > Report Layout > Repeat All Item Labels — needed when you want to use the pivot output as a flat data source.
- **Report Layout:** Compact (default), Outline, Tabular. Tabular puts each field in its own column — closest to a normal table.

6. Pivot Charts, Charts & Dashboards

Chart type selection

Question	Chart
Trend over time	Line (or Area for cumulative)
Compare categories	Column (vertical) or Bar (horizontal, for long labels)
Part-to-whole	Stacked column / 100% stacked; pie only for ≤5 slices
Relationship between two numerics	Scatter
Distribution	Histogram or Box & Whisker
Two different scales	Combo chart with a secondary axis
Progress to target	Bullet-style bar, or gauge

Building a dashboard

The Week 1 deliverable. A repeatable recipe:

1. **Separate your layers.** Sheet 1 = raw data (as a Table). Sheet 2 = pivots and helper calcs. Sheet 3 = the dashboard itself. Never build charts directly on raw data.
2. **Build each pivot** for one specific question. One pivot per chart.
3. **Insert pivot charts** (select pivot > Insert > chart type), then move them to the dashboard sheet (right-click chart > Move Chart > Object in Dashboard).
4. **Add KPI cards:** big single numbers in merged cells, driven by `GETPIVOTDATA` or `SUMIFS`. Format large, bold, with a coloured fill.
5. **Add slicers** and connect them to all pivots via Report Connections, so one click filters the whole dashboard.
6. **Clean up the chrome:** delete gridlines, remove legends where a single series makes them redundant, delete the chart border, set a consistent colour palette, add data labels instead of an axis where it reads better.
7. **Hide the working sheets** and turn off gridlines on the dashboard sheet (View > untick Gridlines).

Conditional formatting

Home > Conditional Formatting: - **Highlight Cells Rules** — greater than, between, text contains, duplicate values - **Top/Bottom Rules** — Top 10 items, Top 10%, above average - **Data Bars / Colour Scales / Icon Sets** — in-cell visuals, good for dense tables - **Use a formula** — the flexible one. Select the range starting at the active cell, then: `=$D2>1000          ' highlight the ENTIRE ROW where column D exceeds 1000   =AND($B2="West", $D2>500)   =COUNTIF($A$2:$A$100, $A2)>1     ' highlight duplicates` Note the mixed reference: `$D2` locks the column so the rule applies across the row, while the row number varies.

Sparklines

Insert > Sparklines > Line/Column/Win-Loss. A tiny in-cell chart per row. Good for a trend column beside a KPI table.

7. Power Query (Get & Transform)

Data > Get Data. The right tool when cleaning must be **repeatable** — the steps are recorded and re-run on refresh, unlike manual edits.

Typical flow: Get Data > From File/Table > the Power Query Editor opens > apply steps > Close & Load.

Common transforms: Remove Columns, Remove Duplicates, Filter Rows, Transform > Format > Trim/Lowercase, Replace Values, Split Column, Change Type, Unpivot Columns (turns a wide cross-tab into long/tidy form — very handy), Merge Queries (a join), Append Queries (a union), Group By.

Every step appears in the Applied Steps pane and can be reordered or deleted. This is the audit trail that manual cleaning lacks.

8. Keyboard Shortcuts

Hands-on assessments are won on speed. Unplug the mouse while drilling.

Navigation & selection

Keys	Action
`Ctrl+Arrow`	Jump to the edge of the data block
`Ctrl+Shift+Arrow`	Select to the edge
`Ctrl+End` / `Ctrl+Home`	Last used cell / A1
`Ctrl+A`	Select the current region
`Ctrl+Space` / `Shift+Space`	Select column / row
`Ctrl+PgUp/PgDn`	Previous / next worksheet
`F5` then Special	Go To Special (blanks, formulas, constants)

Editing

Keys	Action
`F2`	Edit in place
`F4`	Cycle absolute/relative (while editing); repeat last action (otherwise)
`Ctrl+D` / `Ctrl+R`	Fill down / fill right
`Ctrl+Enter`	Fill the entire selection with what you typed
`Alt+Enter`	Line break inside a cell
`Ctrl+;` / `Ctrl+Shift+;`	Insert today’s date / current time
`Ctrl+Shift+V`	Paste Special
`Ctrl+-` / `Ctrl+Shift++`	Delete / insert cells

Formatting & features

Keys	Action
`Ctrl+T`	Create Table
`Ctrl+Shift+L`	Toggle filters
`Alt+=`	AutoSum
`Alt+N+V`	Insert PivotTable
`Alt+H+O+I`	Autofit column width
`Ctrl+1`	Format Cells dialog
`Ctrl+Shift+$ / % / #`	Currency / percent / date format
`Alt+F5`	Refresh pivot
Ctrl+`` ``	Toggle show-formulas view

9. Excel Tables (`Ctrl+T`)

Converting a range to a Table changes several things at once: - **Structured references:** `=SUM(Sales[Revenue])` instead of `=SUM(D2:D5000)`. Readable and self-documenting. - **Auto-expansion:** new rows are included in every formula, chart and pivot that references the table. - Formulas **auto-fill** down the whole column when entered in one cell. - Built-in banded formatting, filter buttons, and an optional Total Row (`Ctrl+Shift+T`).

Structured reference syntax: `Table1[@Revenue]` = this row’s value; `Table1[Revenue]` = the whole column; `Table1[#Headers]`, `Table1[#Totals]`, `Table1[#All]`.

Rename the table under Table Design > Table Name — `Sales` reads far better than `Table1`.

10. Errors and What They Mean

Error	Cause	Typical fix
`#N/A`	Lookup value not found	Check for whitespace/type mismatch; wrap in `IFNA`
`#VALUE!`	Wrong data type in an argument	Text where a number is expected
`#REF!`	Reference to a deleted cell	Rebuild the formula; usually unrecoverable
`#DIV/0!`	Division by zero or blank	`=IFERROR(a/b, 0)`
`#NAME?`	Misspelled function or undefined name	Check spelling; check the range name exists
`#NUM!`	Invalid numeric argument	e.g. `SQRT` of a negative
`#NULL!`	Intersection operator (space) with no overlap	Usually a missing comma between ranges
`#SPILL!`	Dynamic array blocked	Clear the cells in the spill range

Lookup returns `#N/A` but the value is visibly there — almost always one of: trailing space (fix with `TRIM`), number stored as text vs real number, or non-breaking space from a web/PDF paste (fix with `SUBSTITUTE(A2, CHAR(160), "")`).

11. Practice Checklist

Work these until each is under time. Use the practice workbook.

- [ ] Clean a dirty categorical column with `TRIM`/`PROPER` — 2 min
- [ ] Two-table lookup with `INDEX/MATCH` and `IFERROR` — 3 min
- [ ] Multi-criteria `SUMIFS` including a date-range condition — 3 min
- [ ] Build a quarter label from a date and look up a target — 4 min
- [ ] Pivot: Region × Quarter with date grouping and currency format — 5 min
- [ ] Same pivot as % of column total — 1 min
- [ ] Add a calculated field for margin % — 3 min
- [ ] Group a numeric field into buckets — 2 min
- [ ] Top 5 by value filter — 2 min
- [ ] Slicer + timeline connected to two pivots — 5 min
- [ ] Conditional format the top 10% of a column — 2 min
- [ ] Full dashboard: 3 charts + 3 KPI cards + slicers — 25 min